<commit_before>{ stdenv, fetchFromGitHub }:

stdenv.mkDerivation rec {
  name = "libargon2-${version}";
  version = "20171227";

  src = fetchFromGitHub {
    owner = "P-H-C";
    repo = "phc-winner-argon2";
    rev = "${version}";
    sha256 = "0sc9zca1anqk41017vjpas4kxi4cbn0zvicv8vj8p2sb2gy94bh8";
  };

  installPhase = ''
    runHook preInstall
    mkdir -p $out/lib/pkgconfig
    substitute libargon2.pc $out/lib/pkgconfig/libargon2.pc \
      --replace @UPSTREAM_VER@ "${version}"                 \
      --replace @HOST_MULTIARCH@ ""                         \
      --replace 'prefix=/usr' "prefix=$out"

    make install PREFIX=$out
    ln -s $out/lib/libargon2.so $out/lib/libargon2.so.0
    runHook postInstall
  '';

  meta = with stdenv.lib; {
    description = "A key derivation function that was selected as the winner of the Password Hashing Competition in July 2015";
    longDescription = ''
      A password-hashing function created by by <NAME>, <NAME>, and
      <NAME>. Argon2 was declared the winner of the Password
      Hashing Competition (PHC). There were 24 submissions and 9 finalists.
      Catena, Lyra2, Makwa and yescrypt were given special recognition. The PHC
      recommends using Argon2 rather than legacy algorithms.
    '';
    homepage = https://www.argon2.com/;
    license = with licenses; [ asl20 cc0 ];
    maintainers = with maintainers; [ taeer olynch ];
    platforms = platforms.linux ++ platforms.darwin;
  };
}
<commit_msg>libargon2: 20171227 -> 20190702<commit_after>{ stdenv, fetchFromGitHub }:

stdenv.mkDerivation rec {
  pname = "libargon2";
  version = "20190702";

  src = fetchFromGitHub {
    owner = "P-H-C";
    repo = "phc-winner-argon2";
    rev = version;
    sha256 = "0p4ry9dn0mi9js0byijxdyiwx74p1nr8zj7wjpd1fjgqva4sk23i";
  };

  installPhase = ''
    runHook preInstall
    mkdir -p $out/lib/pkgconfig
    substitute libargon2.pc $out/lib/pkgconfig/libargon2.pc \
      --replace @UPSTREAM_VER@ "${version}"                 \
      --replace @HOST_MULTIARCH@ ""                         \
      --replace 'prefix=/usr' "prefix=$out"

    make install PREFIX=$out LIBRARY_REL=lib
    ln -s $out/lib/libargon2.so $out/lib/libargon2.so.0
    runHook postInstall
  '';

  meta = with stdenv.lib; {
    description = "A key derivation function that was selected as the winner of the Password Hashing Competition in July 2015";
    longDescription = ''
      A password-hashing function created by by <NAME>, <NAME>, and
      <NAME>. Argon2 was declared the winner of the Password
      Hashing Competition (PHC). There were 24 submissions and 9 finalists.
      Catena, Lyra2, Makwa and yescrypt were given special recognition. The PHC
      recommends using Argon2 rather than legacy algorithms.
    '';
    homepage = "https://www.argon2.com/";
    license = with licenses; [ asl20 cc0 ];
    maintainers = with maintainers; [ taeer olynch ];
    platforms = platforms.linux ++ platforms.darwin;
  };
}

{'fraction_non_alphanumeric': 0.10879083866621758, 'fraction_numerical': 0.038396766588076794, 'mean_word_length': 2.8125802310654686, 'pattern_counts': {'":': 0, '<': 9, '<?xml version=': 0, '>': 10, 'https://': 2, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2133806', 'n_tokens_mistral': 1057, 'n_tokens_neox': 978, 'n_words': 261}